 1~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~

Public Class frmAudioDevices

    Private Sub frmAudioDevices_Load(sender As Object, e As EventArgs) _
        Handles Me.Load
        txtOutput.Clear()
        'res wird zur Weiterverwendung erzeugt
        Dim res() As String = DeviceEnumerator()

        With txtOutput
            .Rtf = createRTFtext(rtfT)
            .SelectionStart = 0
            .SelectionLength = 0
        End With
    End Sub

End Class

2 ~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~

Imports System.Runtime.InteropServices

Module dhmodDeviceEnumerator

#Region "Vereinbarungen"

    Private Const S_OK As Integer = 0
    Private Const CLSID_MMDeviceEnumerator As String = "bcde0395-e52f-467c-8e3d-c4579291692e"
    Private Const IID_IMMDeviceEnumerator As String = "a95664d2-9614-4f35-a746-de8db63617e6"
    Private Const IID_IMMDeviceCollection As String = "0bd7a1be-7a1a-44db-8397-cc5392387b5e"
    Private Const IID_IMMDevice As String = "d666063f-1587-4e43-81f1-b948e807363f"
    Private Const IID_IPropertyStore As String = "886d8eeb-8cf2-4446-8d02-cdba1dbdcf99"
    'ausschließlich anzuzeigende, spezielle Property, hier Gerätename
    Private Const propDeviceName As String = "b3f8fa53-0004-438e-9003-51a46e139bfc,6"

    Public rtfT() As String 'der zu übergebende RTF-Text
    Dim endNumber As Short = 0, r As Short = -1, rr As Short = -1, activ, equal As Boolean
    Dim resultAudio(), txForm(), store As String, delim As String = ";"

    Private Enum DATA_FLOW As Integer
        Render = 0
        Capture = 1
        All = 2
        DataFlow_Enum_Count = 3
    End Enum

    Private Enum DEVICE_STATE As Integer
        Active = 1
        Disabled = 2
        NotPresent = 4
        Unplugged = 8
    End Enum

    Private Enum STGM_ACCESS As Integer
        Read = 0
        Write = 1
        ReadWrite = 2
    End Enum

    Private Enum Role As Integer
        Console = 0
        Multimedia = 1
        Communications = 2
    End Enum

    Private Enum VARENUM As UShort
        VT_EMPTY = 0
        VT_NULL = 1
        VT_I2 = 2
        VT_I4 = 3
        VT_R4 = 4
        VT_R8 = 5
        VT_CY = 6
        VT_DATE = 7
        VT_BSTR = 8
        VT_DISPATCH = 9
        VT_ERROR = 10
        VT_BOOL = 11
        VT_VARIANT = 12
        VT_UNKNOWN = 13
        VT_DECIMAL = 14
        VT_I1 = 16
        VT_UI1 = 17
        VT_UI2 = 18
        VT_UI4 = 19
        VT_I8 = 20
        VT_UI8 = 21
        VT_INT = 22
        VT_UINT = 23
        VT_VOID = 24
        VT_HRESULT = 25
        VT_PTR = 26
        VT_SAFEARRAY = 27
        VT_CARRAY = 28
        VT_USERDEFINED = 29
        VT_LPSTR = 30
        VT_LPWSTR = 31
        VT_RECORD = 36
        VT_INT_PTR = 37
        VT_UINT_PTR = 38
        VT_FILETIME = 64
        VT_BLOB = 65
        VT_STREAM = 66
        VT_STORAGE = 67
        VT_STREAMED_OBJECT = 68
        VT_STORED_OBJECT = 69
        VT_BLOB_OBJECT = 70
        VT_CF = 71
        VT_CLSID = 72
        VT_VERSIONED_STREAM = 73
        VT_BSTR_BLOB = 4095
        VT_VECTOR = 4096
        VT_ARRAY = 8192
        VT_BYREF = 16384
        VT_RESERVED = 32768
        VT_ILLEGAL = 65535
        VT_ILLEGALMASKED = 4095
        VT_TYPEMASK = 4095
    End Enum

    Private Structure PROPERTYKEY
        Dim fmtid As Guid
        Dim pid As UInteger
    End Structure

    Private Structure CArray
        Dim cElems As UInteger
        Dim pElems As IntPtr
    End Structure

    <StructLayout(LayoutKind.Explicit, Size:=16)> 'gekürzt
    Private Structure PROPVARIANT
        <FieldOffset(0)> Dim vt As VARENUM
        <FieldOffset(2)> Dim wReserved1 As UShort
        <FieldOffset(4)> Dim wReserved2 As UShort
        <FieldOffset(6)> Dim wReserved3 As UShort
        <FieldOffset(8)> Dim bVal As Byte
        <FieldOffset(8)> Dim cVal As SByte
        <FieldOffset(8)> Dim uiVal As UShort
        <FieldOffset(8)> Dim iVal As Short
        <FieldOffset(8)> Dim uintVal As UInteger
        <FieldOffset(8)> Dim intVal As Integer
        <FieldOffset(8)> Dim ulVal As ULong
        <FieldOffset(8)> Dim lVal As Long
        <FieldOffset(8)> Dim fltVal As Single
        <FieldOffset(8)> Dim dblVal As Double
        <FieldOffset(8)> Dim boolVal As Short
        <FieldOffset(8)> Dim pclsidVal As IntPtr
        <FieldOffset(8)> Dim pszVal As IntPtr
        <FieldOffset(8)> Dim pwszVal As IntPtr
        <FieldOffset(8)> Dim punkVal As IntPtr
        <FieldOffset(8)> Dim cArray As CArray
    End Structure
#End Region

#Region "Funktionen"
    <DllImport("Ole32.dll", EntryPoint:="PropVariantClear")>
    <PreserveSig> Private Function PropVariantClear(<[In]> ByRef pvar As PROPVARIANT) As Integer
    End Function

    <DllImport("Propsys.dll", EntryPoint:="PSGetNameFromPropertyKey")>
    <PreserveSig> Private Function PSGetNameFromPropertyKey(<[In]> ByRef propkey As PROPERTYKEY,
                                                                   <Out> ByRef ppszCanonicalName As IntPtr) As Integer
    End Function
#End Region

    ' DeviceEnumeration, speziell für Audio-Geräte
    Public Function DeviceEnumerator() As String()

        setRTFformat()  'spezielle Formate für Erzeugung der RTF-Texte setzen

        Dim MMDeviceEnumerator As IMMDeviceEnumerator _
            = CType(Activator.CreateInstance(Type.GetTypeFromCLSID(New Guid(CLSID_MMDeviceEnumerator))),
              IMMDeviceEnumerator)
        If MMDeviceEnumerator IsNot Nothing Then

            Dim MMDevice2 As IMMDevice = Nothing
            Dim DevIDEP As String = Nothing
            If MMDeviceEnumerator.GetDefaultAudioEndpoint(DATA_FLOW.Render,
                                                          Role.Multimedia, MMDevice2) = S_OK Then
                MMDevice2.GetId(DevIDEP)
            End If

            Dim MMDeviceCollection As IMMDeviceCollection = Nothing
            If MMDeviceEnumerator.EnumAudioEndpoints(DATA_FLOW.Render, DEVICE_STATE.Active,
                                                     MMDeviceCollection) = S_OK Then
                If MMDeviceCollection IsNot Nothing Then
                    Dim Devices As UInteger
                    If MMDeviceCollection.GetCount(Devices) = S_OK Then
                        ' Anzahl der Devices
                        outRTF("Anzahl Audio-Devices: " & Devices & txForm(0))

                        If Devices > 0 Then
                            For Device As UInteger = 0 To Devices - 1UI
                                store = ""
                                Dim MMDevice As IMMDevice = Nothing
                                If MMDeviceCollection.Item(Device, MMDevice) = S_OK Then
                                    If MMDevice IsNot Nothing Then
                                        Dim DevId As String = Nothing
                                        If MMDevice.GetId(DevId) = S_OK Then
                                            'Device Ident-Nummer
                                            outRTF(vbCrLf & StrDup(30, "~") & vbCrLf & "Nummer " _
                                                   & (Device + 1) & vbCrLf _
                                                   & "DEVICE_ID: " & DevId & vbCrLf & txForm(1))
                                        End If

                                        Dim DevState As New DEVICE_STATE
                                        If MMDevice.GetState(DevState) = S_OK Then
                                            'Device Status
                                            outRTF("DEVICE_STATE: " & DevState.ToString & vbCrLf _
                                                   & txForm(2))
                                            'Aktualität
                                            equal = (DevIDEP = DevId)
                                            If equal Then endNumber = Device + 1
                                            outRTF("ACTUALITY: " & equal.ToString & vbCrLf _
                                                   & If(equal, txForm(0), txForm(3)))
                                            store = equal.ToString & delim
                                        End If
                                        Dim PropertyStore As IPropertyStore = Nothing
                                        If MMDevice.OpenPropertyStore(STGM_ACCESS.Read,
                                                                      PropertyStore) = S_OK Then
                                            If PropertyStore IsNot Nothing Then
                                                Dim Props As UInteger
                                                If PropertyStore.GetCount(Props) = S_OK Then
                                                    'Anzahl Properties
                                                    outRTF("Properties exist: " & Props & vbCrLf & txForm(2))
                                                    If Props > 0 Then
                                                        For Prop As UInteger = 0 To Props - 1UI
                                                            Dim PropKey As New PROPERTYKEY
                                                            If PropertyStore.GetAt(Prop, PropKey) = S_OK Then
                                                                ' nur Audio-Gerätenamen ausgeben!!!
                                                                If PropKey.fmtid.ToString & "," & PropKey.pid.ToString = propDeviceName Then
                                                                    'outRTF(vbTab & "PKey: " & PropKey.fmtid.ToString & "," _
                                                                    '       & PropKey.pid.ToString & vbCrLf & txForm(4))

                                                                    'Dim pCanonicalName As IntPtr
                                                                    'If PSGetNameFromPropertyKey(PropKey, pCanonicalName) = S_OK Then
                                                                    '    'Name*
                                                                    '    outRTF(vbTab & "CanonicalName: " & Marshal.PtrToStringUni(pCanonicalName) _
                                                                    '           & vbCrLf & txForm(4))
                                                                    '    Marshal.FreeCoTaskMem(pCanonicalName)
                                                                    'End If

                                                                    Dim PropVar As New PROPVARIANT
                                                                        If PropertyStore.GetValue(PropKey, PropVar) = S_OK Then
                                                                            '
                                                                            outRTF(vbTab & "PropVar.vt: " & PropVar.vt.ToString & vbCrLf & txForm(4))
                                                                            Select Case PropVar.vt
                                                                                Case VARENUM.VT_BOOL
                                                                                    outRTF(vbTab & CBool(PropVar.boolVal).ToString & vbCrLf & txForm(4))
                                                                                Case VARENUM.VT_I2
                                                                                    outRTF(vbCrLf & PropVar.iVal.ToString & vbCrLf & txForm(4))
                                                                                Case VARENUM.VT_I4
                                                                                    outRTF(vbTab & PropVar.intVal.ToString & vbCrLf & txForm(4))
                                                                                Case VARENUM.VT_I8
                                                                                    outRTF(vbTab & PropVar.lVal.ToString & vbCrLf & txForm(4))
                                                                                Case VARENUM.VT_UI2
                                                                                    outRTF(vbTab & PropVar.uiVal.ToString & vbCrLf & txForm(4))
                                                                                Case VARENUM.VT_UI4
                                                                                    outRTF(vbTab & PropVar.uintVal.ToString & vbCrLf & txForm(4))
                                                                                Case VARENUM.VT_UI8
                                                                                    outRTF(vbTab & PropVar.ulVal.ToString & vbCrLf & txForm(3))
                                                                                Case VARENUM.VT_LPWSTR
                                                                                    'der DeviceName
                                                                                    outRTF(vbTab & Marshal.PtrToStringUni(PropVar.pwszVal) & vbCrLf _
                                                                                               & If(equal, txForm(0), txForm(6)))
                                                                                    store += Marshal.PtrToStringUni(PropVar.pwszVal)
                                                                                    rr += 1
                                                                                    ReDim Preserve resultAudio(rr)
                                                                                    resultAudio(rr) = store 'speichern in ResultFeld
                                                                            End Select
                                                                            PropVariantClear(PropVar)
                                                                        End If
                                                                    End If
                                                                End If
                                                        Next Prop
                                                    End If
                                                End If
                                                Marshal.ReleaseComObject(PropertyStore)
                                            End If
                                        End If
                                        Marshal.ReleaseComObject(MMDevice)
                                    End If
                                End If
                            Next Device
                        End If
                    End If
                    Marshal.ReleaseComObject(MMDeviceCollection)
                End If
            End If
            Marshal.ReleaseComObject(MMDeviceEnumerator)
        End If
        Return resultAudio  'Rückgabe des ResultFelds
    End Function

#Region "Interfaces"
    <ComImport>
    <InterfaceType(ComInterfaceType.InterfaceIsIUnknown)>
    <Guid(IID_IMMDeviceEnumerator)>
    Private Interface IMMDeviceEnumerator
        <PreserveSig> Function EnumAudioEndpoints(<[In]> dataFlow As DATA_FLOW,
                                                  <[In]> dwStateMask As DEVICE_STATE,
                                                  <Out, MarshalAs(UnmanagedType.Interface)> ByRef ppDevices As IMMDeviceCollection) As Integer
        <PreserveSig> Function GetDefaultAudioEndpoint(<[In]> dataFlow As DATA_FLOW,
                                                  <[In]> role As Role,
                                                  <Out, MarshalAs(UnmanagedType.Interface)> ByRef ppEndpoint As IMMDevice) As Integer
        <PreserveSig> Function GetDevice() As Integer
        <PreserveSig> Function RegisterEndpointNotificationCallback() As Integer
        <PreserveSig> Function UnregisterEndpointNotificationCallback() As Integer
    End Interface

    <ComImport>
    <InterfaceType(ComInterfaceType.InterfaceIsIUnknown)>
    <Guid(IID_IMMDeviceCollection)>
    Private Interface IMMDeviceCollection
        <PreserveSig> Function GetCount(<Out> ByRef pcDevices As UInteger) As Integer
        <PreserveSig> Function Item(<[In]> nDevice As UInteger,
                                    <Out, MarshalAs(UnmanagedType.Interface)> ByRef ppDevice As IMMDevice) As Integer
    End Interface

    <ComImport>
    <InterfaceType(ComInterfaceType.InterfaceIsIUnknown)>
    <Guid(IID_IMMDevice)>
    Private Interface IMMDevice
        <PreserveSig> Function Activate() As Integer
        <PreserveSig> Function OpenPropertyStore(<[In]> stgmAccess As STGM_ACCESS,
                                                 <Out, MarshalAs(UnmanagedType.Interface)> ByRef ppProperties As IPropertyStore) As Integer
        <PreserveSig> Function GetId(<Out, MarshalAs(UnmanagedType.LPWStr)> ByRef ppstrId As String) As Integer
        <PreserveSig> Function GetState(<Out> ByRef pdwState As DEVICE_STATE) As Integer
    End Interface

    <ComImport>
    <InterfaceType(ComInterfaceType.InterfaceIsIUnknown)>
    <Guid(IID_IPropertyStore)>
    Private Interface IPropertyStore
        <PreserveSig> Function GetCount(<Out> ByRef cProps As UInteger) As Integer
        <PreserveSig> Function GetAt(<[In]> iProp As UInteger,
                                     <Out> ByRef pkey As PROPERTYKEY) As Integer
        <PreserveSig> Function GetValue(<[In]> ByRef pkey As PROPERTYKEY,
                                        <Out> ByRef pv As PROPVARIANT) As Integer
        <PreserveSig> Function SetValue() As Integer
        <PreserveSig> Function Commit() As Integer
    End Interface

#End Region

#Region "Hilfsfunktionen"
    'spezielle Routinen für die Textausgabe

    Sub outRTF(theText As String)
        'die Texte inkl. Formatierungsparameter ins Übergabe-Feld speichern
        r += 1
        ReDim Preserve rtfT(r)
        rtfT(r) = theText
    End Sub

    Sub setRTFformat()
        'definieren verschiedener Formate für RTF-Text
        txForm =
            {";LightPink;Black;Arial,16,Bold", ";LightCyan;Black;Arial,12,Regular",
            ";Bisque;Black;Arial,12,Regular", ";Wheat;Black;Arial,12,Bold",
            ";SeaShell;Black;Arial,12,Regular", ";LightPink;Black;Arial,12,Regular",
            ";Wheat;Black;Arial,12,Regular"}
    End Sub
#End Region

End Module

3 ~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~

    ''' <summary>
    ''' 	RTF-Text mit Formatierungen erstellen
    ''' </summary>
    ''' <param name="tFeld">das Feld einzelner, zu formatierender Texte</param>
    ''' <param name="defBGColor">die Hintergrundfarbe, default</param>
    ''' <returns>der formatierte RTF-Text</returns>
    Public Function createRTFtext(tFeld() As String,
                 Optional defBGColor As Color = Nothing) As String
        ' Aufbau eines Feld-Elements der Texte
        ' Definition: text, selectionback, selectionfor, selectionfont
        ' Beispiel für Feld-Element: "Hallo Welt;Black;White;Arial,12,Regular"
        Dim rtt As New RichTextBox
        Dim i, j As Short
        Dim sp(), spf(), t As String
        Dim cb, cf As Color
        Dim fo As Font, fs As FontStyle
        rtt.Clear()
        ' das Array verarbeiten
        For i = 0 To tFeld.Length - 1
            If IsNothing(tFeld(i)) Then Continue For
            ' Feldelement splitten
            sp = Split(tFeld(i), ";")
            ' gesplittetes Element verarbeiten
            For j = 0 To sp.Length - 1
                Select Case j
                    Case 0   ' der Text
                        t = sp(0)
                    Case 1   ' SelectionBackColor
                        If IsNothing(defBGColor) OrElse sp(1) <> "" Then
                            cb = Color.FromName(sp(1))
                        ElseIf Not IsNothing(defBGColor) And sp(1) = "" Then
                            cb = defBGColor
                        End If
                    Case 2   ' SelectionColor
                        cf = Color.FromName(sp(2))
                    Case 3   ' der Font
                        ' splitten der Font-Parameter 
                        spf = Split(sp(3), ",")
                        ' festlegen des FontStyles
                        Select Case spf(2)
                            Case "Regular" : fs = FontStyle.Regular
                            Case "Bold" : fs = FontStyle.Bold
                            Case "Italic" : fs = FontStyle.Italic
                            Case "Underline" : fs = FontStyle.Underline
                        End Select
                        ' den Font definieren
                        fo = New Font(spf(0), CShort(spf(1)), fs)
                End Select
            Next
            ' den RTF-Text in einer RTF-Textbox erzeugen oder nur als Zeile
            With rtt
                If Not IsNothing(cb) Then .SelectionBackColor = cb
                If Not IsNothing(cf) Then .SelectionColor = cf
                If Not IsNothing(fo) Then .SelectionFont = fo
                .AppendText(t)
            End With
            cb = Nothing : cf = Nothing : fo = Nothing
        Next
        ' Rückgabe des RTF-Textes
        Return rtt.Rtf
    End Function



